

```

gap> g:= SymmetricGroup( 4 );
Sym( [ 1 .. 4 ] )
gap> tbl:= CharacterTable( g );; HasIrr( tbl );
i5 : betti(t,Weights=>{1,0})
false
      0 1 2 3 4 gap> tblmod2:= CharacterTable( tbl, 2 );
o5 = total: 1 4 13 14 4 BrauerTable( Sym( [ 1 .. 4 ] ), 2 )
      0: 1 . . . .
      1: . 2 2 4 2 gap> tblmod2 = CharacterTable( tbl, 2 );
      2: . 2 5 6 . true
      3: . . 4 . 2
      4: . . . 4 . gap> tblmod2 = BrauerTable( tbl, 2 );
      5: . . 2 . . true
gap> tblmod2 = BrauerTable( tbl, 2 );
o5 : BrauerTable( Sym( [ 1 .. 4 ] ), 2 )
i6 : betti(t,Weights=>{0,1})
      0 1 2 3 4 gap> libtbl:= CharacterTable( "M" );
o6 = total: 1 4 13 14 4 CharacterTable( "M" )
      0: 1 . . . . gap> CharacterTableRegular( libtbl, 2 );
      1: . 2 2 . 2 BrauerTable( "M", 2 );
      2: . 2 5 6 . BrauerTable( libtbl, 2 );
      3: . . 4 . 2 gap> BrauerTable( libtbl, 2 );
      4: . . . 4 . fail
      5: . . 2 . .
gap> CharacterTable( "Symmetric", 4 );
o6 : BettiTally CharacterTable( "Sym(4)" )
i7 : t1 = betti(t,Weights=>{1,1})
gap> ComputedBrauerTables( tbl );
      0 1 2 3 4 [ , BrauerTable( Sym( [ 1 .. 4 ] ), 2 ) ]
o7 = total: 1 4 13 14 4
      0: 1 . . . .
      1: . . . . .
      2: . . . . .
      3: . 2 . . .
      4: . . . . .
      5: . 2 . . .
      6: . . 1 . .
      7: . . 8 6 .
      8: . . 4 8 4
o7 : BettiTally
i8 : peek t1
      ring r1 = 32003,(x,y,z),ds;
      int a,b,c,t=11,5,3,0;
      poly f = x^a+y^b+z^(3*c)+x^(c+2)*y^(c-1)+x^(
      x^(c-2)*y^c*(y^2+t*x)^2;
      option(noprot);
      timer=1;
      ring r2 = 32003,(x,y,z),dp;
      poly f=imap(r1,f);
      ideal j=jacob(f);
      vdim(std(j));
==> 536
      vdim(std(j+f));
==> 195
      timer=0; // reset timer
o8 = BettiTally{(0, {0, 0}, 0) => 1 }
      (1, {2, 2}, 4) => 2
      (1, {3, 3}, 6) => 2
      (2, {3, 7}, 10) => 2
      (2, {4, 4}, 8) => 1
      (2, {4, 5}, 9) => 4
      (2, {5, 4}, 9) => 4
      (2, {7, 3}, 10) => 2
      (3, {4, 7}, 11) => 4
      (3, {5, 5}, 10) => 6
      (4, {5, 7}, 12) => 2
      (4, {7, 5}, 12) => 2

```

# Journal of Software for Algebra and Geometry

The Geometric Decomposability package for Macaulay2  
MIKE CUMMINGS AND ADAM VAN TUYL



# The GeometricDecomposability package for Macaulay2

MIKE CUMMINGS AND ADAM VAN TUYL

**ABSTRACT:** Using the geometric vertex decomposition property, first defined by Knutson, Miller, and Yong, a recursive definition for geometrically vertex decomposable ideals was given by Klein and Rajchgot. We introduce the *Macaulay2* package `GeometricDecomposability` which provides a suite of tools to experiment and test the geometric vertex decomposability property of an ideal.

**1. INTRODUCTION.** The *geometric vertex decomposition* of an ideal was first introduced by Knutson, Miller, and Yong [9] as part of their study of vexillary matrix Schubert varieties. Geometric vertex decomposition can be viewed as a generalization of a vertex decomposition of a simplicial complex. Using the notion of a geometric vertex decomposition, Klein and Rajchgot [7] introduced *geometrically vertex decomposable ideals*. These ideals, which are defined recursively, were partially inspired by the definition of a vertex decomposable simplicial complex, a recursively defined family of simplicial complexes.

As shown by both [9] and [7], ideals that have a geometric vertex decomposition, or are geometrically vertex decomposable, have other desirable properties. As one such example, Klein and Rajchgot [7, Corollary 4.8] have shown that homogeneous geometrically vertex decomposable ideals are *glicci*, i.e., these ideals belong to the Gorenstein liaison class of a complete intersection. Further properties of geometrically vertex decomposable ideals have been developed in [4; 6; 8; 5].

Due to their recent introduction, there are many features of geometrically vertex decomposable ideals that are still not known. To facilitate further experimentation and exploration, we have created `GeometricDecomposability`, a package for *Macaulay2* that enables researchers to test and search for ideals that are geometrically vertex decomposable. In particular, our package allows the user to check if a given ideal satisfies the geometric vertex decomposition property of [9] or the geometrically vertex decomposable property (or its variants) as found in [7]. This note reviews the needed mathematical background, summarizes the main features of our packages, and provides some illustrative examples.

**2. MATHEMATICAL BACKGROUND.** We summarize the mathematical background to define the geometric vertex decomposition property and geometrically vertex decomposable ideals. Throughout,  $k$  denotes a field.

---

*MSC2020:* primary 13P10; secondary 14M25.

*Keywords:* geometric vertex decomposition, geometrically vertex decomposable, Macaulay2.

`GeometricDecomposability` version 1.2

Let  $y$  be a variable of the polynomial ring  $R = k[x_1, \dots, x_n]$ . A monomial ordering  $<$  on  $R$  is said to be  $y$ -compatible if the initial term of  $f$  satisfies  $\text{in}_<(f) = \text{in}_<(\text{in}_y(f))$  for all  $f \in R$ . Here,  $\text{in}_y(f)$  is the *initial  $y$ -form* of  $f$ , that is, if  $f = \sum_i \alpha_i y^i$  and  $\alpha_d \neq 0$  but  $\alpha_t = 0$  for all  $t > d$ , then  $\text{in}_y(f) = \alpha_d y^d$ . We set  $\text{in}_y(I) = \langle \text{in}_y(f) \mid f \in I \rangle$  to be the ideal generated by all the initial  $y$ -forms of an ideal  $I$  in  $R$ .

Given an ideal  $I \subseteq R$  and a  $y$ -compatible monomial ordering  $<$ , let  $G(I) = \{g_1, \dots, g_m\}$  be a Gröbner basis of  $I$  with respect to this ordering. For  $i = 1, \dots, m$ , write  $g_i$  as  $g_i = q_i y^{d_i} + r_i$ , where  $y$  does not divide any term of  $q_i$  and  $\text{in}_y(g_i) = q_i y^{d_i}$ . This second condition is equivalent to no term of  $r_i$  is divisible by  $y^{d_i}$ . Given this setup, we define the two ideals

$$C_{y,I} = \langle q_1, \dots, q_m \rangle \quad \text{and} \quad N_{y,I} = \langle q_i \mid d_i = 0 \rangle.$$

Following Knutson, Miller, and Yong [9], we make the following definition:

**Definition 2.1** [9, Section 2]. With the notation as above, the ideal  $I$  has a *geometric vertex decomposition with respect to  $y$*  if

$$\text{in}_y(I) = C_{y,I} \cap (N_{y,I} + \langle y \rangle).$$

Using Definition 2.1, Klein and Rajchgot [7] recursively defined geometrically vertex decomposable ideals. Recall that an ideal  $I$  is *unmixed* if all of its associated primes have the same height.

**Definition 2.2** [7, Definition 2.7]. An ideal  $I$  of  $R = k[x_1, \dots, x_n]$  is *geometrically vertex decomposable* if  $I$  is unmixed and

- (1)  $I = \langle 1 \rangle$ , or  $I$  is generated by a (possibly empty) subset of variables of  $R$ , or
- (2) there is a variable  $y = x_i$  in  $R$  and a  $y$ -compatible monomial ordering  $<$  such that  $I$  has a geometric vertex decomposition with respect to  $y$ , i.e.,

$$\text{in}_y(I) = C_{y,I} \cap (N_{y,I} + \langle y \rangle),$$

and the contractions of the ideals  $C_{y,I}$  and  $N_{y,I}$  to the ring  $k[x_1, \dots, \hat{y}, \dots, x_n]$  are geometrically vertex decomposable.

The ideals  $\langle 0 \rangle$  and  $\langle 1 \rangle$  in the ring  $k$  are also considered geometrically vertex decomposable by convention.

Klein and Rajchgot also introduced two variants on the geometrically vertex decomposable property. We also recall these definitions. For the first variant, observe that in the Definition 2.2, you do not need to use the induced monomial order for the contractions. Indeed, it could be the case that you need to pick different monomial orders to verify that  $C_{y,I}$  and  $N_{y,I}$  are geometrically vertex decomposable. For  $<$ -compatibly geometrically vertex decomposable ideals, a fixed lexicographical order (and its induced monomial orders) are used throughout; the formal definition for this class of ideals is given below.

**Definition 2.3** [7, Definition 2.11]. Fix a lexicographical order  $<$  on  $R = k[x_1, \dots, x_n]$ . An ideal  $I \subseteq R$  is  $<$ -compatibly *geometrically vertex decomposable* if  $I$  is unmixed and

- (1)  $I = \langle 1 \rangle$ , or  $I$  is generated by a (possibly empty) subset of variables of  $R$ , or

- (2) for the largest variable  $y = x_i$  in  $R$  with respect to the order  $<$ , the ideal  $I$  has a geometric vertex decomposition with respect to  $y$ , and the contractions of the ideals  $C_{y,I}$  and  $N_{y,I}$  to the ring  $S = k[x_1, \dots, \hat{y}, \dots, x_n]$  are  $<$ -compatible geometrically vertex decomposable, where we use  $<$  to also denote the natural induced monomial order on  $S$ .

The second variant relaxes some conditions on the ideals  $C_{y,I}$  and  $N_{y,I}$ , giving a weaker version of geometrically vertex decomposable ideals. The definition was inspired by Nagel and Römer’s notion of a weakly vertex decomposable simplicial complex; see [10, Definition 3.1]. Following [7, Section 2], the geometric vertex decomposition

$$\text{in}_y(I) = C_{y,I} \cap (N_{y,I} + \langle y \rangle)$$

is *degenerate* if  $C_{y,I} = \langle 1 \rangle$  or if  $\sqrt{C_{y,I}} = \sqrt{N_{y,I}}$ , and it is *nondegenerate* otherwise. Note that the definition requires the field to be infinite.

**Definition 2.4** [7, Definition 4.6]. Let  $k$  be an infinite field. An ideal  $I \subseteq R = k[x_1, \dots, x_n]$  is *weakly geometrically vertex decomposable* if  $I$  is unmixed and

- (1)  $I = \langle 1 \rangle$ , or  $I$  is generated by a (possibly empty) subset of variables of  $R$ , or
- (2) (degenerate case) for some variable  $y = x_j$  of  $R$ ,  $\text{in}_y(I) = C_{y,I} \cap (N_{y,I} + \langle y \rangle)$  is a degenerate geometric vertex decomposition and the contraction of  $N_{y,I}$  to the ring  $k[x_1, \dots, \hat{y}, \dots, x_n]$  is weakly geometrically vertex decomposable, or
- (3) (nondegenerate case) for some variable  $y = x_j$  of  $R$ ,  $\text{in}_y(I) = C_{y,I} \cap (N_{y,I} + \langle y \rangle)$  is a nondegenerate geometric vertex decomposition, the contraction of  $C_{y,I}$  to the ring  $k[x_1, \dots, \hat{y}, \dots, x_n]$  is weakly geometrically vertex decomposable, and  $N_{y,I}$  is radical and Cohen–Macaulay.

Properties of geometrically vertex decomposable ideals,  $<$ -compatibly geometrically vertex decomposable ideals, and weakly geometrically vertex decomposable ideals were developed in [7]. In particular, it was shown that these ideals can give insight into questions related to liaison theory.

**3. THE PACKAGE.** The *Macaulay2* package `GeometricDecomposability` was created as a tool to determine whether an ideal  $I$  in  $R = k[x_1, \dots, x_n]$  is geometrically vertex decomposable (or if it satisfies one of the variants). We highlight some of the key features of this package.

Our package is built around the function `oneStepGVD`, which is designed to test whether or not an ideal  $I$  has a geometric vertex decomposition with respect to a given variable  $y$ . In other words, this function determines if a given  $I$  and  $y$  satisfy the properties of Definition 2.1. As seen from Definitions 2.2, 2.3, and 2.4, checking whether or not an ideal has a geometric vertex decomposition is key step in these recursive definitions. Our choice of name for this function was motivated by the fact that this function allows us to move one iteration, or “one step”, in the recursive definition.

For a given variable  $y$ , the function `oneStepGVD` returns a sequence, where the first element in the sequence is true or false depending on whether or not the given variable  $y$  gives a geometric vertex

decomposition of  $I$ , while the second element is the ideal  $C_{y,I}$ , and the third element is the ideal  $N_{y,I}$ . As an illustration, we consider the ideal found in [7, Example 2.16]:

```
i1 : loadPackage "GeometricDecomposability";
i2 : R = QQ[a..f];
i3 : I = ideal(b*(c*f - a^2), b*d*e, d*e*(c^2+a*c+d*e+f^2));
i4 : oneStepGVD(I,b)
o4 = (true, ideal (a*c*d*e + c^2 d*e + d^2 e^2 + d*e*f^2, d*e, a^2 - c*f),
-----
ideal(a*c*d*e + c^2 d*e + d^2 e^2 + d*e*f^2))
```

In this case, we do have a geometric vertex decomposition. If, on the other hand, we asked if the ideal has a geometric vertex decomposition with respect to the variable  $c$ , we get a negative answer:

```
i5 : oneStepGVD(I,c)
o5 = (false, ideal (b*d*e, b*f, a*d*e + d*e), ideal(b*d*e))
```

We want to highlight that the ideals  $C_{y,I}$  and  $N_{y,I}$  do not depend upon the choice of the Gröbner basis or a particular  $y$ -compatible order (see the comment after [7, Definition 2.3]). In our package, when we compute  $C_{y,I}$  and  $N_{y,I}$ , we use a lexicographical ordering on  $R$  where  $y > x_j$  for all  $i \neq j$  if  $y = x_i$  since this gives us an easily accessible  $y$ -compatible order.

If the user only requires the ideal  $C_{y,I}$  or  $N_{y,I}$ , we have built functions to find these ideals, namely `CyI` and `NyI`. These functions actually call `oneStepGVD`, and then return the relevant item in the list. Continuing with the example above, we have:

```
i6 : CyI(I,c)
o6 = ideal (b*d*e, b*f, a*d*e + d*e)
```

Note that this is the second entry in the sequence in output `o5`.

As a tool to encourage experimentation, we have also included the function `findOneStepGVD`. Given an ideal  $I$  in  $R = k[x_1, \dots, x_n]$ , it returns a list of all the variables with which  $I$  has a geometric vertex decomposition. In our running example, there is only one such variable:

```
i7 : findOneStepGVD(I)
o7 = {b}
```

We have created three separate functions to check if an ideal is geometrically vertex decomposable,  $<$ -compatibly geometrically vertex decomposable, or weakly geometrically vertex decomposable. The implementation of each function requires repeated use of the function `oneStepGVD`.

To show that our running example is geometrically vertex decomposable, we enter the command:

```
i8 : isGVD(I)
o8 = true
```

Running the above command with the option `isGVD(I,Verbose=>True)` will allow the user to identify which variable is used to form the geometric vertex decomposition at each step.

Using the `isLexCompatiblyGVD` command, we can check if an ideal is  $\prec$ -compatibly geometrically vertex decomposable. The user must specify the ideal, and the specific lexicographical order by providing an ordering of the variables. As an example, we can check if our running example is  $\prec$ -compatibly geometrically vertex decomposable with respect to the lexicographical order which satisfies  $f > e > d > c > b > a$ . The specific command is:

```
i9 : isLexCompatiblyGVD(I,{f,e,d,c,b,a})
o9 = false
```

We can also search over all possible lexicographical orders to determine if an ideal is  $\prec$ -compatibly geometrically vertex decomposable. Specifically, the command `findLexCompatiblyGVDOrders` returns all the lexicographical orders for which the ideal is  $\prec$ -compatibly geometrically vertex decomposable. In our example, there is no such lexicographical order, as given by the output:

```
i10 : findLexCompatiblyGVDOrders(I)
o10 = {}
```

This agrees with [7, Example 2.16] which proved that this ideal is not  $\prec$ -compatibly geometrically vertex decomposable. Note that running this command can be computationally expensive since one may need to check  $n!$  different lexicographical orders in  $k[x_1, \dots, x_n]$ .

Finally, the command `isWeaklyGVD` enables the user to check if an ideal satisfies Definition 2.4. By [7, Corollary 4.7], all geometrically vertex decomposable ideals are weakly geometrically vertex decomposable. Our running example is therefore weakly geometrically vertex decomposable, as expected:

```
i11 : isWeaklyGVD(I)
o11 = true
```

There are ideals that are weakly geometrically vertex decomposable, but not geometrically vertex decomposable. The following example comes from [7, Example 4.10]:

```
i12 : J = ideal(b*(c*f-a^2),b*d*e,d*e*(a^2+f^2+d*e));
i13 : isWeaklyGVD(J)
o13 = true
i14 : isGVD(J)
o14 = false
```

We present a new example of an ideal with this property at the end of this paper.

**4. EXAMPLES.** We illustrate the `GeometricDecomposability` package using examples from square-free monomial ideals and toric ideals of graphs.

***Squarefree monomial ideals.*** As noted in the introduction, geometrically vertex decomposable ideals was inspired by the definition of vertex decomposable simplicial complexes. We flesh out this connection, and explain the output of package within the context of squarefree monomial ideals.

Let  $V = \{x_1, \dots, x_n\}$  be a vertex set, and let  $2^V$  denote the power set of  $V$ . A *simplicial complex* is a subset  $\Delta \subseteq 2^V$  that satisfies the two properties: (i) if  $F \in \Delta$  and  $G \subseteq F$ , then  $G \in \Delta$ , and (ii)  $\{x_i\} \in \Delta$  for

all  $i \in \{1, \dots, n\}$ . The maximal elements of  $\Delta$  with respect to inclusion are called the *facets*. A simplicial complex is *pure* if all the facets have the same cardinality. Given a vertex  $x \in V$ , the *deletion* of  $x$  is the subcomplex

$$\text{del}_\Delta(x) = \{F \in \Delta \mid F \cap \{x\} = \emptyset\}$$

and the *link* of  $x$  is

$$\text{link}_\Delta(x) = \{F \mid F \cap \{x\} = \emptyset \text{ and } F \cup \{x\} \in \Delta\}.$$

Note that the link and deletion are not necessarily simplicial complexes on  $V$ , but on a subset of  $V$ . Precisely,  $\text{link}_\Delta(x)$  is a simplicial complex on  $\bigcup_{F \in \text{link}_\Delta(x)} F$  and  $\text{del}_\Delta(x)$  is a simplicial complex on  $\bigcup_{F \in \text{del}_\Delta(x)} F$ .

Vertex decomposable simplicial complexes were first introduced by Provan and Billera [11]:

**Definition 4.1.** A simplicial complex  $\Delta$  on  $V$  is *vertex decomposable* if  $\Delta$  is pure and either (i)  $\Delta = \emptyset$ , or  $\Delta$  is a simplex, i.e., the only facet is  $\{x_1, \dots, x_n\}$ , or (ii), there exists a vertex  $x \in V$  such that  $\text{del}_\Delta(x)$  and  $\text{link}_\Delta(x)$  are vertex decomposable.

To connect Definition 4.1 with the definition of geometrically vertex decomposable ideals from Definition 2.2, we use the Stanley–Reisner correspondence. In particular, given a simplicial complex  $\Delta$ , define

$$I_\Delta = \langle x_{i_1} \cdots x_{i_j} \mid \{x_{i_1}, \dots, x_{i_j}\} \notin \Delta \rangle \subseteq R = k[x_1, \dots, x_n]$$

to be the squarefree monomial ideal generated by monomials corresponding to subsets not in  $\Delta$ . The connection between the two definitions then comes via the following theorem:

**Theorem 4.2** [7, Proposition 2.9]. *Let  $\Delta$  be a simplicial complex on  $V = \{x_1, \dots, x_n\}$ . Then  $\Delta$  is vertex decomposable if and only if  $I_\Delta$  is geometrically vertex decomposable.*

In other words, the above theorem says that the squarefree monomial ideals that are geometrically vertex decomposable are precisely those squarefree monomial ideals that are the Stanley–Reisner ideals of a vertex decomposable simplicial complex.

Our example below highlights the connection between the  $\text{link}_\Delta(x)$  and  $\text{del}_\Delta(x)$  and the ideals  $C_{y,I}$  and  $N_{y,I}$  that appear in the definition of (weakly) geometrically vertex decomposable ideals. Starting a fresh *Macaulay2* session, consider the simplicial complex  $\Delta$  on the vertex set  $\{a, b, c, d, e\}$  with facets  $\{\{a, c\}, \{a, d\}, \{b, d\}, \{b, e\}, \{c, e\}\}$ . Using the `SimplicialDecomposability` package [1], we can input this simplicial complex and check that it is vertex decomposable:

```
i1 : loadPackage "GeometricDecomposability";
i2 : loadPackage "SimplicialDecomposability";
i3 : R = QQ[a..e];
i4 : Delta = simplicialComplex {a*c,a*d,b*d,b*e,c*e};
i5 : isVertexDecomposable(Delta)
o5 = true
```



We can now obtain the simplicial complexes  $\text{del}_\Delta(a)$  and  $\text{link}_\Delta(a)$  of the vertex  $a \in \{a, \dots, e\}$  using the following commands. We also include the corresponding Stanley–Reisner ideal of each simplicial complex:

```
i6 : Link = link(Delta,a);
i7 : Delete = faceDelete(a,Delta);
i8 : IDelta = monomialIdeal Delta
o8 = monomialIdeal (a*b, b*c, c*d, d*e, e*a)
i9 : monomialIdeal Link
o9 = monomialIdeal (a, b, c*d, e)
i10 : monomialIdeal Delete
o10 = monomialIdeal (a, b*c, c*d, d*e)
```

Now consider the output of the `oneStepGVD` command with input  $I_\Delta$  and the vertex  $a$ :

```
i11 : oneStepGVD(IDelta,a)
o11 = (true, ideal (d*e, c*d, b*c, e, b), ideal (d*e, c*d, b*c))
```

If we compare the ideals  $C_{a,I_\Delta}$  and  $N_{a,I_\Delta}$ , the second and third ideals in the above list, with the Stanley–Reisner ideals of  $\text{link}_\Delta(a)$  and  $\text{del}_\Delta(a)$ , they are the same except that the later ideals have an extra generator, namely the variable  $a$ . (Note that the given generators of  $C_{a,I_\Delta}$  are not a minimal set of generators). Technically, the simplicial complexes  $\text{link}_\Delta(a)$  and  $\text{del}_\Delta(a)$  are simplicial complexes on the vertex set  $\{b, c, d, e\}$ , and so their corresponding Stanley–Reisner ideals should belong to  $k[b, c, d, e]$ . We can move all the ideals to this ring, and then we verify that we have an equality of ideals:

```
i12 : S = QQ[b,c,d,e]
i13 : C1=substitute(CyI(IDelta,a),S)
o13 = ideal (d*e, c*d, b*c, e, b)
i14 : N1=substitute(NyI(IDelta,a),S)
o14 = ideal (d*e, c*d, b*c)
i15 : L1=substitute(monomialIdeal Link,S)
o15 = ideal (0, b, c*d, e)
i16 : D1=substitute(monomialIdeal Delete,S)
o16 = ideal (0, b*c, c*d, d*e)
```

In general, the ideal  $C_{y,I}$ , respectively the ideal  $N_{y,I}$ , can be viewed as the algebraic analog of the link of a vertex, respectively the deletion of a vertex, of a simplicial complex.

As a final calculation, we verify that our Stanley–Reisner ideal is geometrically vertex decomposable, as expected by Theorem 4.2:

```
i17 : isGVD(IDelta)
o17 = true
```

**Toric ideals of graphs.** For our second example, we use our package to find minimal examples of toric ideals of graphs that are (weakly) geometrically vertex decomposable.

Let  $G = (V, E)$  be a finite simple graph with vertex set  $V = \{x_1, \dots, x_n\}$  and edge set  $E = \{e_1, \dots, e_m\}$ . Let  $R = k[e_1, \dots, e_m]$  and  $S = k[x_1, \dots, x_n]$ . We define a map  $\varphi : R \rightarrow S$  by  $\varphi(e_i) = x_j x_k$ ,

where  $e_i = \{x_j, x_k\}$ . The *toric ideal of  $G$* , denoted  $I_G$ , is the kernel of this map, that is,  $I_G = \ker \varphi$ . It can be shown (see, for example [13, Proposition 10.1.5]) that  $I_G$  is prime binomial ideal. Furthermore, the generators of  $I_G$  correspond to closed even walks in the graph  $G$ . Informally, a closed even walk is a sequence of adjacent edges that start and stop at the same edge (see [13, Chapter 7.1] for the formal definition). Not every graph has a closed even walk, e.g., trees, so in some cases  $I_G = \langle 0 \rangle$ .

Geometrically vertex decomposable toric ideals of graphs were first studied in [4]. It was shown that every toric ideal of a bipartite graph is geometrically vertex decomposable. However, not every toric ideal of a graph is geometrically vertex decomposable, as noted in [4, Remark 7.1].

Using our package `GeometricDecomposability` we can find minimal examples of graphs that are (weakly) geometric vertex decomposable. By [4, Theorem 3.3], we can restrict our search to connected graphs. Table 1 summarizes the results of our computation. For all connected graphs with  $e$  edges with  $e \in \{4, \dots, 9\}$ , we first constructed the toric ideal  $I_G$ . If  $I_G \neq \langle 0 \rangle$ , we then checked if the ideal was geometrically vertex decomposable and weakly geometrically vertex decomposable. The second column of Table 1 is the number of simple connected graphs on  $e$  edges (this is sequence A002905 in [12]), the third column is the number of such graphs that have a nonzero toric ideal, while the fourth and fifth record the number of these ideals that are geometrically vertex decomposable (GVD) or weakly GVD.

This data was computed also using the *Macaulay2* packages `Nauty` and `FourTiTwo` [3; 2]. We used the following code:


```
loadPackage "GeometricDecomposability";
loadPackage "NautyGraphs";
loadPackage "FourTiTwo";

getToricIdeal = (A,R) -> (
  -- A, an incidence matrix; R, a ring
  m = product gens R;
  return saturate(sub(toBinomial(transpose(syz(A)),R),R),m);
)

R = QQ[a..i]; -- ring with 9 indeterminates (for the up to 9 edges we will see)
-- all graphs with number of edges between 4 and 9 (inclusive)
GList = flatten for n from 4 to 10 list generateGraphs(n, 4, 9, OnlyConnected=>true);
-- for each number of edges, filter the list to these graphs, and check if GVD
for E from 4 to 9 do (
  f := buildGraphFilter {"NumEdges" => E};
  HList := filterGraphs(GList, f);
  print("There are " | toString (#HList) | " graphs on " | toString E | " edges,");
  -- create lists of graphs, incidence matrices of these graphs, and toric ideals
  -- of these graphs
  graphList = for g in HList list stringToGraph g;
  incMatList = for G in graphList list incidenceMatrix G;
  idealList = for A in incMatList list getToricIdeal A;
  -- look at the subset of the ideals which are not the zero ideal, weakly GVD, and GVD
  nonZeroList = select(idealList, i-> i != 0);
  print(toString(#nonZeroList) | " of which are nonzero");
  wgvdList = select(nonZeroList, i -> isWeaklyGVD(i));
  print(toString(#wgvdList) | " of which are weakly GVD and");
  gvdList = select(nonZeroList, i -> isGVD(i));
  print(toString(#gvdList) | " of which are GVD");
  print("");
);
```

edges	simple connected graphs	nonzero toric ideals	weakly GVD toric ideals	GVD toric ideals
4	5	1	1	1
5	12	2	2	2
6	30	11	11	11
7	79	33	33	33
8	227	125	124	124
9	710	449	445	444

**Table 1.** Comparison of the number of (weakly) GVD toric ideals of graphs.



**Figure 1.** The unique graph  $G$  with 8 edges whose toric ideal is not GVD (left), and the unique graph  $H$  with 9 edges whose toric ideal is weakly GVD but not GVD (right).

Table 1 implies that there is exactly one graph on 8 edges (and hence the smallest graph) whose toric ideal is not geometrically vertex decomposable. Figure 1 shows the unique graph  $G$  on 8 edges whose toric ideal is not geometrically vertex decomposable. Note that this graph is the same graph of [4, Remark 7.1]. The toric ideal of this graph is  $I_G = \langle ad^2fg - bce^2h \rangle$ .

Our computations also imply there is a unique graph on 9 edges whose toric ideal is weakly geometrically vertex decomposable, but not geometrically vertex decomposable (and moreover, this is the smallest such graph). This graph  $H$  appears to the right in Figure 1. The toric ideal of this graph is  $I_H = \langle fg - ei, bcef - ad^2g \rangle$ .

**ACKNOWLEDGMENTS.** We thank Sergio Da Silva, Patrica Klein, and Jenna Rajchgot for comments and suggestions. We additionally thank the referees for concrete suggestions that improved both this manuscript and the software package. Cummings was partially supported by the Milos Novotny Fellowship and an NSERC USRA and CGS-M. Van Tuyl’s research is partially supported by the NSERC Discovery Grant 2019-05412.

**SUPPLEMENT.** The online supplement contains version 1.2 of GeometricDecomposability.

#### REFERENCES.

- [1] D. Cook, II, “Simplicial decomposability”, *J. Softw. Algebra Geom.* **2:1** (2010), 20–23. MR Zbl
- [2] D. Cook, II, “Nauty in Macaulay2”, *J. Softw. Algebra Geom.* **3:1** (2011), 1–4. MR Zbl
- [3] D. Cook, II, “Nauty: interface to nauty”, available at <https://github.com/Macaulay2/M2/tree/master/M2/Macaulay2/packages>. (Macaulay2 package) version 1.4.3.1.

- [4] M. Cummings, S. Da Silva, J. Rajchgot, and A. Van Tuyl, “Geometric vertex decomposition and liaison for toric ideals of graphs”, *Algebr. Comb.* **6:4** (2023), 965–997. MR Zbl
- [5] S. Da Silva and M. Harada, “Geometric vertex decomposition, Gröbner bases, and Frobenius splittings for regular nilpotent Hessenberg varieties”, *Transform. Groups* (online publication July 2023). Zbl
- [6] P. Klein, “Diagonal degenerations of matrix Schubert varieties”, *Algebr. Comb.* **6:4** (2023), 1073–1094. MR Zbl
- [7] P. Klein and J. Rajchgot, “Geometric vertex decomposition and liaison”, *Forum Math. Sigma* **9** (2021), art. id. e70. MR Zbl
- [8] P. Klein and A. Weigandt, “Bumpless pipe dreams encode Gröbner geometry of Schubert polynomials”, preprint, 2021. arXiv 2108.08370
- [9] A. Knutson, E. Miller, and A. Yong, “Gröbner geometry of vertex decompositions and of flagged tableaux”, *J. Reine Angew. Math.* **630** (2009), 1–31. MR Zbl
- [10] U. Nagel and T. Römer, “Glicci simplicial complexes”, *J. Pure Appl. Algebra* **212:10** (2008), 2250–2258. MR Zbl
- [11] J. S. Provan and L. J. Billera, “Decompositions of simplicial complexes related to diameters of convex polyhedra”, *Math. Oper. Res.* **5:4** (1980), 576–594. MR Zbl
- [12] N. J. A. Sloane, “A002905: Number of connected graphs with  $n$  edges”, entry A002905 in *The On-Line Encyclopedia of Integer Sequences* (<http://oeis.org>), 2022. Zbl
- [13] R. H. Villarreal, *Monomial algebras*, 2nd ed., CRC Press, Boca Raton, FL, 2015. MR Zbl

RECEIVED: 10 Nov 2022

REVISED: 17 Oct 2023

ACCEPTED: 23 Jan 2024

MIKE CUMMINGS:

cummin5@mcmaster.ca

Department of Mathematics and Statistics, McMaster University, Hamilton, ON, Canada

ADAM VAN TUYL:

vantuyl@math.mcmaster.ca

Department of Mathematics and Statistics, McMaster University, Hamilton, ON, Canada